UNIVERSIDAD PONTIFICIA DE SALAMANCA

Ampliación de Matemáticas, Curso 2005/06

Preparado por: **Lic. Raúl Martín Martín**

Práctica 3

En esta segunda práctica tratamos los siguientes temas:

- Representación de señales discretas y continuas
- Cálculo diferencial e integral en una y varias variables
- Repaso de las series de Fourier y cálculo de sus coeficientes
- Introducción a las Ecuaciones Diferenciales

Práctica I:

EJERCICIO 1.- Dibuja los siguientes casos

1) Coseno desplazado: cos(t), cos(t+2), cos(t-2)
2) Exponencial por coseno: $\exp(2t)\cdot\cos(2\pi t+\pi/4)$
3) Exponenciales discretas armónicamente relacionadas: $\exp(ik\Omega_o n)$
4) Exponencial compleja: $\mathrm{Re}(\exp(it\pi/3))$ y $\mathrm{Im}(\exp(it\pi/3))$
5) Coseno discreto: cos(n/5), cos(nPi/5), cos(2n)

Las prácticas se guardarán en vuestra unidad Z, en un subdirectorio llamado **maple**. Para esta segunda práctica se creará un subdirectorio en maple llamado **practica2**. Dentro de este subdirectorio, se guardaran los archivos mws utilizando la siguiente nomenclatura:

Maple2*a*_primer apellido_segundo apellido.mws

donde *a* es el número de la práctica.

Es obligatorio respetar estos nombres para los ficheros y subdirectorios.

Utiliza la almohadilla, #, para los comentarios. Cuando no quieras que se muestre el resultado de una instrucción, finalízala con : y cuando sí quieras que se visualice, con ;

Para dibujar las señales discretas, utiliza una secuencia de órdenes del tipo:

```
> p:=[[n,cos(2*n)] $n=-20..20]:
plot(p, style=point,symbol=circle,title="cos(2*n)");
```

Para definir una función x(t):

```
> x:=t->exp(2*t)*(cos(2*Pi*t+(Pi/4))):
```

luego puedes dibujarla utilizando la orden display

```
p:=plot(x(t),t=0..15/8,legend=…):
display(p);
```

Aquí tienes las gráficas que debes obtener:
















t
Legend
Re(exp(j·Pi·t/3))
Im(exp(j·Pi·t/3))







Práctica II. Derivación e integración en una variable.

Definición.- Sea f una función real de dominio D y S el subconjunto de D formado por los puntos en los que f es diferenciable. Se llama <u>derivada</u> de f a la función f' tal que a cada x $\in$ S asocia la derivada de f en x, es decir:

$$f'(x) = \lim_{h \to 0} \frac{f(x+h) - f(x)}{h}$$

Como la derivada de una función es ora función, podremos tratar de hallar su derivada. Se lleva derivada segunda de f a la derivada de f' y se representa por f''. Es decir, f''(s)=(f')'(x). También notamos la derivada segunda de alguna de las siguientes formas:

$$D^2 f, \frac{d^2 f}{dx^2}, \frac{d^2 y}{dx^2} \quad o \quad y''$$

EJERCICIO 2.- Obtener las derivadas primera y segunda de:

$$\frac{\tan(x^2)}{x+\cos x}$$

Definición.- Sea f una función real definida en el intervalo [a,b], continua. El límite I de las sumas de Riemann correspondientes a f se llama **integral definida** o **integral de Riemann** de f en el intervalo [a,b] y se representa por el símbolo:

$$I = \int_a^b f(x)dx.$$

La variable x se llama variable de integración. La función f se conoce como integrando y los extremos a y b como límites de integración, inferior y superior respectivamente.

EJERCICIO 3.-

1. Estimar el área limitada por la gráfica de la función $f(x) = x^2 + 1$ y el eje OX entre a=0 y b=1 utilizando n=5 rectángulos de la misma longitud y de altura el valor de f(x) en el punto medio de dichos intervalos.
2. Repetir el apartado anterior para n=10,20,…,50.
3. Determinar una fórmula general que proporcione una estimación de dicha área mediante n rectángulos de las características anteriores.
4. Calcular el límite cuando n tiene a + ∞ de la expresión anterior.
5. Comprobar directamente que la integral definida de f(x) en [a,b] es precisamente el límite anterior.

(**Indicación**: Cargar los paquetes student y plots, y usar las funciones middlebox y middlesum).

<u>Cálculo de una integral indefinida</u>

Calcular $\int \frac{x}{x^3+1}dx.$

```
> integrando:=x/(x^3+1);
```

$$integrando := \frac{x}{x^3+1}$$

```
> int(integrando,x);
```

$$-\frac{1}{3}\ln(x+1)+\frac{1}{6}\ln(x^2-x+1)+\frac{1}{3}\sqrt{3}\,\arctan\left(\frac{1}{3}(2x-1)\sqrt{3}\right)$$

Práctica III. Derivadas parciales

Ya hemos definido en clase qué entendemos por una función real de n-variables.

Representación gráfica.-

Definición.- Sea f(x,y) una función real de dos variables y domino D. Se llama **gráfica** de f al conjunto de puntos (x,y,z) de R^3 tales que z = f(x,y), con (x,y) $\in$ D.

EJERCICIO 4.- Sea f(x,y) = 4- $x^2 - y^2$.

a) Representar gráficamente la superficie z=f(x,y) para valores de (x,y) pertenecientes al recinto:

$$D = \{(x,y)\in\Re^2 \,/\, -1 \le x \le 1,\ -1 \le y \le 1\}.$$

b) Representar conjuntamente la superficie anterior y el dominio D, éste último en el plano z=0.

Derivadas parciales.

Definición.- Sea f(x,y) una función real de dos variables con domino D abierto. Se llama derivada parcial respecto de x en (a,b) al límite siguiente (cuando exista):

$$\frac{\partial f}{\partial x}(a,b) = \lim_{h\to 0}\frac{f(a+h,b)-f(a,b)}{h}$$

Análogamente, se llama derivada parcial respecto de y en (a,b) al límite

$$\frac{\partial f}{\partial y}(a,b) = \lim_{k\to 0}\frac{f(a,b+k)-f(a,b)}{k}$$

La derivada parcial respecto de x en (a,b) representa:

1. La razón de cambio de la variable dependiente z=f(x,y) cuando x varía e y se mantiene constante.
2. La pendiente de la recta tangente a la curva del plano y=b obtenida como intersección de la superficie z=f(x,y) con dicho plano.

De manera análoga, la derivada parcial respecto de y en (a,b) representa:

3. La razón de cambio de la variable dependiente z=f(x,y) cuando y varía y x se mantiene constante.
4. La pendiente de la recta tangente a la curva del plano x=a obtenida como intersección de la superficie z=f(x,y) con dicho plano.

EJERCICIO 5.- Sea $f(x,y) = 100 - 20\,x^2 - 30\,y^2$.

a) Calcular las derivadas parciales (respecto de x e y) en el punto P=(1,1).

b) Dibujar:

I. La superficie z = f(x,y)

II. La curva intersección de la superficie anterior con el plano y=1. (Representa esta curva en paramétricas)

III. La recta del plano y=1 que pasa por el punto (1,1,f(1,1)) y cuya pendiente es $f_x(1,1)$.

c) Dibujar:

I. La superficie z = f(x,y)

II. La curva intersección de la superficie anterior con el plano x=1.

III. La recta del plano x=1 que pasa por el punto (1,1,f(1,1)) y cuya pendiente es $f_y(1,1)$.

(**Indicación.-** Para representar varias curvas en un espacio de 3 dimensiones, usar el comando spacecurve)

Práctica IV. Series de Fourier.

Un comando para dar el desarrollo de una serie es:

```
> series(exp(x),x);
```

$$1+x+\frac{1}{2}x^2+\frac{1}{6}x^3+\frac{1}{24}x^4+\frac{1}{120}x^5+\mathrm{O}(x^6)$$

El primer argumento indica la función que queremos desarrollar y el segundo la variable. También podemos dar un tercer argumento que indica el orden de la expansión:

```
> series(exp(x),x,10);
```

$$1+x+\frac{1}{2}x^2+\frac{1}{6}x^3+\frac{1}{24}x^4+\frac{1}{120}x^5+\frac{1}{720}x^6+\frac{1}{5040}x^7+\frac{1}{40320}x^8+\frac{1}{362880}x^9+\mathrm{O}(x^{10})$$

Cuando no se da el tercer argumento, Maple usa por defecto 6.

También podemos obtener un desarrollo de Taylor para funciones de varias variables:

```
> mtaylor(sin(x^2+y^2),[x,y],8);
```

$$x^2+y^2-\frac{1}{6}x^6-\frac{1}{2}y^2x^4-\frac{1}{2}y^4x^2-\frac{1}{6}y^6$$

Pero a nosotros en este curso, nos interesan las **series de Fourier**

Recuerda que la serie de Fourier de una función f(x) viene definida por:

$$f(x)=\frac{a_0}{2}+\sum_{n=1}^{\infty}\left(a_n\cos(nx)+b_n\sin(nx)\right)$$

donde

$$a_n=\frac{1}{\pi}\int_{-\pi}^{\pi}f(x)\cos(nx)dx$$

$$b_n=\frac{1}{\pi}\int_{-\pi}^{\pi}f(x)\sin(nx)dx$$

EJERCICIO 6. Desarrolla un programa para calcular las series de Fourier. (De forma que al darle la función nos devuelva la serie de Fourier y después dibuja la función y su aproximación).

PRÁCTICA V.

Ecuaciones diferenciales ordinarias

El paquete DEtools. Carga el paquete anterior con la orden **with(DEtools);** Puedes ver los comandos especiales que posee para resolver ODE's.

Las derivadas primera, segunda, tercera… las podemos calcular escribiendo

diff(y(x), x); diff(y(x),x,x); diff(y(x), x, x, x)…

Otra forma en la que podemos calcular derivadas, es usando el comando D, escribiendo D(y)(x), (D@@2)(y)(x), (D@@3)(y)(x)…

```
D(sin@(2*x));
```

$$2\,(\cos@(2\,x))\,\mathrm{D}(x)$$

El comando más usa para saber el comportamiento de las ecuaciones diferenciales ordinarias (EDOs) (ODEs, en inglés) es el comando **dsolve**. La sintaxis es:

dsolve(eqns, vars)

Escribimos algunas EDOs con y sin valores iniciales, observa y prueba los ejemplos:

```
> eq:=diff(v(t),t)+2*t=0;
```

$$eq := \left(\frac{\partial}{\partial t}\mathrm{v}(t)\right) + 2\,t = 0$$

```
> ini:=v(1)=5;
```

$$ini := \mathrm{v}(1) = 5$$

```
> dsolve({eq,ini},{v(t)});
```

$$\mathrm{v}(t) = -t^2 + 6$$

```
> restart;
> eq:=diff(y(x),x$2)-y(x)=1;
```

$$eq := \left(\frac{\partial^2}{\partial x^2}\mathrm{y}(x)\right) - \mathrm{y}(x) = 1$$

```
> dsolve({eq},{y(x)});
```

$$\{\mathrm{y}(x) = \mathrm{e}^{x}\,\_C2 + \mathrm{e}^{(-x)}\,\_C1 - 1\}$$

```
> restart;
>
> de1:=diff(y(t),t$2)+5*diff(y(t),t)+6*y(t)=0;
```

$$de1 := \left(\frac{\partial^2}{\partial t^2}\mathrm{y}(t)\right) + 5\left(\frac{\partial}{\partial t}\mathrm{y}(t)\right) + 6\,\mathrm{y}(t) = 0$$

```
> ini:=y(0)=0,D(y)(0)=1;
```

$$ini := \mathrm{y}(0) = 0,\ \mathrm{D}(y)(0) = 1$$

```
> dsolve({de1,ini},{y(t)});
```

$$\mathrm{y}(t) = \mathrm{e}^{(-2\,t)} - \mathrm{e}^{(-3\,t)}$$

La opción type=numeric. Consideremos la ecuación diferencial y una condición inicial:

```
> eq:=x(t)*diff(x(t),t)=t^2;
```

$$eq := \mathrm{x}(t)\left(\frac{\partial}{\partial t}\mathrm{x}(t)\right) = t^2$$

```
> ini:=x(1)=2;
```

$$ini := \mathrm{x}(1) = 2$$

El resultado del comando dsolve con la opción numérica es un procedimiento que devuelve una lista de ecuaciones:

```
> sol:=dsolve({eq,ini},{x(t)},type=numeric);
```

$$sol := \mathbf{proc}(rkf45\_x)\ \ldots\ \mathbf{end\ proc}$$

La solución satisface la condición inicial:

```
> sol(1);
```

$$[t = 1.,\ \mathrm{x}(t) = 2.]$$

```
> sol(0);
```

$$[t = 0.,\ \mathrm{x}(t) = 1.82574365940448402]$$

Usamos el comando **eval** para seleccionar un valor particular de la lista de ecuaciones:

```
> eval(x(t),sol(1));
```

$$2.$$

```
> eval([t,x(t)],sol(0));
```

$$[0.,\ 1.82574365940448402]$$

El paquete plots, contiene el comando **odeplot**. Trabaja con él. ¿Qué características tiene?

```
> with(plots):
> odeplot(sol,[t,x(t)], -1..2);
```



(**odeplot** dibuja el resultado de dsolve(…., type=numeric).

Ver **?plots, odeplot** para la sintaxis de **odeplot**.

Dibujamos Ecuaciones diferenciales ordinarias.-

Sabemos que no podemos resolver muchas ecuaciones diferenciales de manera exacta. En tales casos, el dibujo de la ecuación diferencial, nos ayuda a "vislumbrar" cual es la solución:

```
> ode1:=
> diff(y(t), t$2) + sin(t)^2*diff(y(t),t)+y(t)=cos(t)^2;
```

$$ode1 := \left(\frac{\partial^2}{\partial t^2}\mathrm{y}(t)\right) + \sin(t)^2\left(\frac{\partial}{\partial t}\mathrm{y}(t)\right) + \mathrm{y}(t) = \cos(t)^2$$

```
> ic1:=y(0)=1, D(y)(0)=0;
```

$$ic1 := \mathrm{y}(0) = 1, \mathrm{D}(y)(0) = 0$$

Resuelve la ecuación con **dsolve**

```
> dsolve({ode1,ic1},{y(t)});
```

¿Qué ocurre?

Intentamos encontrar el comando **DEplot** en el paquete DEtools. Lo cargamos:

```
> with(DEtools):
```

DEplot tiene la siguiente sintaxis:

DEplot(EDO, variable dependiente, rango, [condiciones iniciales])

En el caso de arriba:

```
> DEplot(ode1, y(t), 0..20, [[ic1]]);
```



Podemos « refinar » el dibujo especificando un paso más pequeño:

```
> DEplot(ode1,y(t),0..20,[[ic1]],stepsize=0.2);
```



Si queremos especificar mas que una lista de condiciones iniciales, DEplot dibuja una solución para cada lista:

```
> ic2:=y(0)=0, D(y)(0)=1;
```

$$ic2 := \mathrm{y}(0) = 0, \mathrm{D}(y)(0) = 1$$

```
> DEplot(ode1,y(t),0..20,[[ic1],[ic2]], stepsize=0.2);
```



La función de Heaviside.-

Consideremos

```
> eq:=diff(y(t),t)=-y(t)*Heaviside(t-1);
```

$$eq := \frac{\partial}{\partial t}\, y(t) = -y(t)\, \mathrm{Heaviside}(t-1)$$

```
> ini:=y(0)=3;
```

$$ini := y(0) = 3$$

```
> dsolve({eq,ini},{y(t)});
```

$$y(t) = 3\, e^{((-t+1)\, \mathrm{Heaviside}(t-1))}$$

Convertimos la solución a una función que se pueda dibujar:

```
> rhs(%);
```

$$3\, e^{((-t+1)\, \mathrm{Heaviside}(t-1))}$$

```
> f:=unapply(%,t);
```

$$f := t \to 3\, e^{((-t+1)\, \mathrm{Heaviside}(t-1))}$$

```
> plot(f,0..4);
```



Resolvemos la misma ecuación numéricamente:

```
> sol1:=dsolve({eq,ini},{y(t)}, type=numeric);
```

$$sol1 := \mathrm{proc}(rkf45\_x) \ \ldots \ \mathrm{end\ proc}$$

Usamos el comando odeplot:

```
> odeplot(sol1,[t,y(t)],0..4);
```



La función delta de Dirac.- Podemos usar esta función de manera similar a la función de Heaviside para producir impulsos:

```
> eq:=diff(y(t),t)=-y(t)*Dirac(t-1);
```

$$eq := \frac{\partial}{\partial t}\,\mathrm{y}(t) = -\mathrm{y}(t)\,\mathrm{Dirac}(t-1)$$

```
> ini:=y(0)=3;
```

$$ini := \mathrm{y}(0) = 3$$

```
> dsolve({eq,ini},{y(t)});
```

$$\mathrm{y}(t) = 3\,\mathrm{e}^{(-\mathrm{Heaviside}(t-1))}$$

Convertimos la solución a una función que se pueda dibujar:

```
> rhs(%);
```

$$3\,\mathrm{e}^{(-\mathrm{Heaviside}(t-1))}$$

```
> f:=unapply(%,t);
```

$$f := t \to 3\,\mathrm{e}^{(-\mathrm{Heaviside}(t-1))}$$

```
> plot(f,0..4);
```



Funciones a trozos.- Ya vimos como se definian:

```
> f:=x->piecewise(1<=x and x<2, 1, 0);
```

$$f := x \rightarrow \text{piecewise}(1 \le x \text{ and } x < 2, 1, 0)$$

```
> f(x);
```

$$\begin{cases} 1 & 1 - x \le 0 \text{ and } x < 2 \\ 0 & \textit{otherwise} \end{cases}$$

```
> plot(f,0..3);
```



Podemos hacer tambien:

```
> eq:=diff(y(t),t)=1-y(t)*f(t);
```

$$eq := \frac{\partial}{\partial t} \mathrm{y}(t) = 1 - \mathrm{y}(t) \left(\begin{cases} 1 & 1 - t \le 0 \text{ and } t < 2 \\ 0 & \textit{otherwise} \end{cases} \right)$$

```
> ini:=y(0)=3;
```

$$ini := \mathrm{y}(0) = 3$$

```
> sol3:=dsolve({eq,ini},{y(t)}, type=numeric);
```

$$sol3 := \mathbf{proc}(rkf45\_x) \ \ldots \ \mathbf{end\ proc}$$

De nuevo, usando el comando **odeplot** para dibujar el resultado:

```
> with(plots, odeplot):
odeplot(sol3,[t,y(t)],0..4);
```



EJERCICIO 7.- Resolver la ecuación

$$y' = y.\frac{y+2x-1}{x+y}$$

Dibujar las soluciones en un mismo gráfico tomando como valores para las constantes C=1,-1,1/10,-1/10. Usa distintos colores y leyendas.

EJERCICIO 8.- La temperatura de un cuerpo está regida por la ecuación diferencial $\frac{dT}{dt}+kT=kQ(t)$, siendo Q(t) la temperatura ambiente y k una constante positiva. Para un cuerpo determinado, k=0,25 cuando T se mide en grados Celsius y el tiempo en horas.

a) Si Q(t)=10º y T(0)=20º, ¿Qué temperatura alcanza el cuerpo a las 3 horas ?

b) Si Q(t)=15+0.2t y T(0)=12º, ¿cuánto tiempo es necesario para que la temperatura del cuerpo difiera en 1º de la temperatura ambiente ?

EJERCICIO 9.- Denomine L(t) la longitud de un pez en el instante t y suponga que el pez crece de acuerdo con la ecuación de crecimiento restringido:

$$\frac{dL}{dt} = k(34 - L(t)) \qquad con \quad L(0) = 2$$

(a) Resuelva la ecuación anterior

(b) Utilice la solución de (a) para determinar k bajo el supuesto de que L(4)=10. Dibuje la gráfica de L(t) para este valor de k.

(c) Calcule la longitud del pez cuanto t=10.

(d) Calcule la longitud asintótica del pez, es decir, calcule $\lim_{t \to \infty} L(t)$.

ALGORITMO PARA METODO DE EULER

```
[tmetodo,xmetodo]=mi_metodo(f,intervalo,x0,paso)
                          ^
                          | |       | |_tamaño de paso
                          | |       |_dato inicial
                          | |_intervalo en forma []
                          |_función
```

DATOS (Entrada)

```
f(t,x)              % función de la EDO
tint                % intervalo de existencia
x0                  % valor inicial
h                   % paso


SALIDA

tmetodo                 % tabla de puntos t(i) en COLUMNAS
xmetodo                 % tabla de valores de x(t) en COLUMNAS

PASO 1              % construcción de variables auxiliares

tinic               % tiempo inicial
tfin                % tiempo final
N = (tfin - tinic)/h    % número de intervalos

tmetodo=[tinic]         % inicializa la lista
wexmetodo=[x0]              % inicializa la lista



PASO 2

z= x0               % inicializa variable bucle
s=tinic             % inicializa variable bucle

for i=1:N

z= z+ h* f(s,z)         % siguiente valor,
                    % actualiza variable bucle
s=s+h                   % siguiente valor
                    % actualiza variable bucle

tmetodo=[tmetodo,s]             % acumula en la lista
xmetodo=[xmetodo,z]              % acumula en la lista

end
```

EJERCICIO 4.- Programar el método de Euler.

Usar para ello la ecuación y’(x)=2xy(x), con la condición inicial x(0)=0, y(0)=1 y siendo h=1/10, considerando el intervalo [x_0, x_0+2], siendo N+1 el número de puntos de la partición. Comparar, gráficamente, la solución numérica con la solución exacta.

Ejercicio.- La temperatura de un cuerpo está regida por la ecuación diferencial $\frac{dT}{dt} + kT = kQ(t)$, siendo Q(t) la temperatura ambiente y k una constante positiva. Para un cuerpo determinado, k=0,25 cuando T se mide en grados Celsius y el tiempo en horas.

c) Si Q(t)=10º y T(0)=20º, ¿Qué temperatura alcanza el cuerpo a las 3 horas ?

d) Si Q(t)=15+0.2t y T(0)=12º, ¿cuánto tiempo es necesario para que la temperatura del cuerpo difiera en 1º de la temperatura ambiente ?

IMPLEMENTANDO CON MAPLE.-

```
> restart;
with(DEtools):
k:=0.25:
Q(t):=10:
ec:=diff(T(t),t)+k*T(t)-k*Q(t):
sol:=dsolve({ec,T(0)=20}):
sol;
DEplot(ec=0,T(t),t=0..10,[[T(0)=20]],T=0..20,arrows=NONE);
```

$$T(t) = 10 + 10\, e^{(-1/4\, t)}$$




```
> #resuelve la EDO y la dibuja para Q(t) no constante.
restart;
with(DEtools):
k:=0.25:
Q(t):=15 + t/5:
ec:=diff(T(t),t)+k*T(t)-k*Q(t):
Tsol:=dsolve({ec,T(0)=12}):
Tsol;
eval(Tsol,t=3);
evalf(%);
DEplot(ec=0,T(t),t=0..30,[[T(0)=12]],arrows=NONE,title="Q(t)=15+
t/5 To=12 k=1/4");

T(t):=71/5+t/5-11*exp(-t/4)/5:
```

```
rect(t):=71/5+t/5:
plot([Q(t),T(t),rect(t)],t=0..20,title="Q(t)=15+t/5 To=12 k=1/4
rectas x=71/5+t/5 y Q(t)");
print("tiempo en el que la temperatura ambiente y la del objeto
difieren en 1 grado Celsius:");
a:=solve(Q(t)-T(t)=1,t);
evalf(a);
```

$$T(t) = \frac{71}{5} + \frac{1}{5}t - \frac{11}{5}e^{(-1/4\,t)}$$

$$T(3) = \frac{74}{5} - \frac{11}{5}e^{(-3/4)}$$

$$T(3) = 13.76079358$$





"tiempo en el que la temperatura ambiente y la del objeto difieren en 1 grado Celsius:"

$$a := 4\ln(11)$$

$$9.591581092$$